Circuit activ de ordin I – derivator

Scopul lucrarii

In lucrarea de fata ne propunem analiza unui derivator cu AO cu condensator si rezistor in bucla de reactie.



Fig.1 Derivator



Fig.2 Circuit practic

Obs: -> In Fig1. se prezinta schema unui derivator cu AO, ideal.
->In Fig2. se prezinta schema realizata practic, numita in continuare circuit.

Caracterizarea circuitului

```
> restart:with(Syrup):
> libname:="c:\\maple/SCSlib",libname:
```

Circuit real cu rezistenta serie

Descrierea circuitului

```
> circuit:=
"Derivator cu AO
Vg 1 0 Vg
r 1 2 r
C 2 3 C
R 3 4 R
E 4 0 0 3 A
.end":
```

Calculul tensiunilor nodale si a curentilor prin laturi

```
> syrup(circuit,ac,'curenti','tensiuni');
Syrup/parsedeck:   Analyzing SPICE deck "Derivator cu AO" (ignoring this
line)
```

$$\{v_1 = Vg, v_4 = -\frac{s\,C\,R\,A\,Vg}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r}, v_2 = \frac{Vg\,(s\,C\,R + A + 1)}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r},$$
$$v_3 = \frac{s\,C\,R\,Vg}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r}\}$$

```
> tensiuni;
```

$$\{v_1 = Vg, v_4 = -\frac{s\,C\,R\,A\,Vg}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r}, v_2 = \frac{Vg\,(s\,C\,R + A + 1)}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r},$$
$$v_3 = \frac{s\,C\,R\,Vg}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r}\}$$

```
> curenti;
```

$$\left\{ i_E = \frac{s\,C\,(A+1)\,Vg}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r}, i_{Vg} = -\frac{s\,C\,(A+1)\,Vg}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r},\right.$$
$$i_r = \frac{Vg - \frac{Vg\,(s\,C\,R + A + 1)}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r}}{r}, i_C = \frac{s\,C\,(A+1)\,Vg}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r},$$
$$\left. i_R = \frac{\frac{s\,C\,R\,Vg}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r} + \frac{s\,C\,R\,A\,Vg}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r}}{R} \right\}$$

Calcularea functiei de transfer

Exista mai multe metode de calcul a functiei de transfer:

<u>Metoda 1</u>: folosind formula amplificarii

Stim ca functia de transfer a unui asemenea circuit este descrisa de formula:

$$H(s) = -\frac{Z_2(s)}{Z_1(s)},\ Z_1(s) = r + \frac{1}{sC},\ Z_2(s) = R$$

Astfel functia de transfer devine:

$$H(s) = -\frac{R}{r + \frac{1}{sC}} = -\frac{s \cdot \frac{RC}{rC}}{s + \frac{1}{rC}} = -\frac{RC}{rC} \cdot \frac{s}{s + \frac{1}{rC}}$$

Notand: $a = \frac{1}{RC}$ si $a_1 = \frac{1}{rC}$ va rezulta ca:

$$H(s) = -\frac{a_1}{a} * \frac{s}{s + a_1}$$

<u>Metoda 2</u>: folosind modelul nulor-norator

Cu ajutorul modelului nulor-norator al AO:



Fig .3 Modelul nulor norator

Scriind ecuatiile Kirchoff 1 si 2 pe circuit:

$$\begin{cases} V_{in} = Z_1 I \\ V_{out} = Z_2 I \end{cases} \quad \begin{cases} V_{in} = (r + \frac{1}{sC}) I \\ V_{out} = RI \end{cases}$$

Obtinem aceeasi functie de transfer obtinuta cu metoda 1, Z1 si Z2 avand aceeasi forma.

<u>Metoda 3</u>: calcul simbolic

Folosind Maple si pachetul Syrup se obtine:

```
> Hcircuit:=eval(v[4]/v[1],tensiuni);
```

$$Hcircuit := -\frac{s\,C\,R\,A}{s\,C\,R + A + 1 + A\,s\,C\,r + s\,C\,r}$$

Calcularea functiei de transfer pentru AO ideal

```
> Hcircuitideal:=limit(Hcircuit,A=infinity);
```

$$Hcircuitideal := -\frac{s\,C\,R}{1 + s\,C\,r}$$

Descrierea folosind ecuatiile de stare a circuitului

1.Ecuatii de stare

```
> syrup(circuit,tran,'curenti','tensiuni');
Syrup/parsedeck:   Analyzing SPICE deck "Derivator cu AO" (ignoring this
line)
```

$$\{v_C(0) = 0, \frac{\partial}{\partial t} v_C(t) = \frac{A\,Vg + Vg - A\,v_C(t) - v_C(t)}{C\,(R + r + A\,r)}\}, \{v_C(t)\}$$

2. Ecuatii de iesire

```
> tensiuni;
```

$$\{v_2 = \frac{R\,Vg + r\,A\,v_C(t) + r\,v_C(t)}{R + r + A\,r}, v_4 = -\frac{A\,R\,(Vg - v_C(t))}{R + r + A\,r}, v_3 = \frac{R\,(Vg - v_C(t))}{R + r + A\,r}, v_1 = Vg\}$$

```
> curenti;
```

$$\{i_{Vg} = -\frac{A\,Vg + Vg - A\,v_C(t) - v_C(t)}{R + r + A\,r}, i_r = \frac{Vg - \frac{R\,Vg + r\,A\,v_C(t) + r\,v_C(t)}{R + r + A\,r}}{r},$$

$$i_C = \frac{A\,Vg + Vg - A\,v_C(t) - v_C(t)}{R + r + A\,r},\ i_R = \frac{\dfrac{R\,(Vg - v_C(t))}{R + r + A\,r} + \dfrac{A\,R\,(Vg - v_C(t))}{R + r + A\,r}}{R},$$

$$\left. i_E = \frac{A\,Vg + Vg - A\,v_C(t) - v_C(t)}{R + r + A\,r} \right\}$$

Derivator

Descrierea circuitului

```
> derivator:=
"Derivator cu AO
Vin 1 0 Vin
R 2 3 R
C 1 2 C
E 3 0 0 2 A
.end":
```

Calculul tensiunilor nodale si a curentilor prin laturi

```
> syrup(derivator,ac,'curenti','tensiuni'):
Syrup/parsedeck:   Analyzing SPICE deck "Derivator cu AO" (ignoring this
line)
> tensiuni;
```

$$\left\{ v_1 = Vin,\ v_3 = -\frac{A\,Vin\,s\,C\,R}{s\,C\,R + A + 1},\ v_2 = \frac{Vin\,s\,C\,R}{s\,C\,R + A + 1} \right\}$$

```
> curenti;
```

$$\left\{ i_C = \frac{s\,C\,Vin\,(A+1)}{s\,C\,R + A + 1},\ i_R = \frac{\dfrac{Vin\,s\,C\,R}{s\,C\,R + A + 1} + \dfrac{A\,Vin\,s\,C\,R}{s\,C\,R + A + 1}}{R},\ i_E = \frac{s\,C\,Vin\,(A+1)}{s\,C\,R + A + 1},\right.$$

$$\left. i_{Vin} = -\frac{s\,C\,Vin\,(A+1)}{s\,C\,R + A + 1} \right\}$$

Calcularea functiei de transfer pentru circuit

Exista mai multe metode de calcul a functiei de transfer:

<u>Metoda 1</u>: folosind formula amplificarii

Stim ca functia de transfer a unui asemenea circuit este descrisa de formula:

$$H(s) = -\frac{Z_2(s)}{Z_1(s)},\ Z_1(s) = \frac{1}{s*C},\ Z_2(s) = G$$

Astfel functia de transfer devine:

$$H(s) = -sRC$$

<u>Metoda 2</u>: folosind modelul nulor-norator

Cu ajutorul modelului nulor-norator al AO:



Fig .3 Modelul nulor norator

Scriind ecuatiile Kirchoff 1 si 2 pe circuit obtinem:

$$\begin{cases} V_{in} = Z_1 I \\ V_{out} = Z_2 I \end{cases} \quad \begin{cases} V_{in} = \dfrac{1}{sC} I \\ V_{out} = RI \end{cases}$$

aceeasi functie de transfer obtinuta cu metoda 1, Z1 si Z2 avand aceeasi forma.

Metoda 3: calcul simbolic

Folosind Maple si pachetul Syrup se obtine:

```
> H:=eval(v[3]/v[1],tensiuni);
```

$$H := -\frac{A\, s\, C\, R}{s\, C\, R + A + 1}$$

Calcularea functiei de transfer pentru AO ideal

```
> Hideal:=limit(H,A=infinity);
```

$$Hideal := -s\, C\, R$$

Descrierea folosind ecuatiile de stare a circuitului

1.Ecuatii de stare

```
> syrup(derivator,tran,'curenti','tensiuni');
Syrup/parsedeck:   Analyzing SPICE deck "Derivator cu AO" (ignoring this line)
```

$$\{ v_C(0) = 0, \frac{\partial}{\partial t} v_C(t) = \frac{Vin - v_C(t) + A\, Vin - A\, v_C(t)}{C\, R} \}, \{ v_C(t) \}$$

2. Ecuatii de iesire

```
> tensiuni;
```

$$\{ v_3 = -A\, Vin + A\, v_C(t), v_1 = Vin, v_2 = Vin - v_C(t) \}$$

```
> curenti;
```

$$\{ i_{Vin} = -\frac{Vin - v_C(t) + A\, Vin - A\, v_C(t)}{R}, i_E = \frac{Vin - v_C(t) + A\, Vin - A\, v_C(t)}{R},$$
$$i_R = \frac{Vin - v_C(t) + A\, Vin - A\, v_C(t)}{R}, i_C = \frac{Vin - v_C(t) + A\, Vin - A\, v_C(t)}{R} \}$$

Pentru circuitul real daca consideram o gama de frecvente pentru care rezistenta R1 >>Zc(s):

```
> limit(Hcircuitideal,r=0);
```

$$-s\, C\, R$$

Obs: calculind functia de transfer pentru circuitul real in gama de frecventa pentru care rezistenta r se poate neglija s-a obtinut aceeasi relatie.

Functia de transfer H(s)

S-a calculat in sectiunea anterioara functia de transfer pentru circuitul real (cu rezistenta r) si pentru derivator:

$$\text{derivator: } H(s) = -\frac{s}{a}$$

$$\text{circuit: } H(s) = -\frac{a_1}{a} * \frac{s}{s + a_1} \text{ cu } a_1 = \frac{1}{rC},\ a = \frac{1}{RC}$$

```
> Hideal;Hcircuitideal;
```

$$-s\, C\, R$$

$$-\frac{s\, C\, R}{1 + s\, C\, r}$$

Polii functiei de transfer:

```
> RootOf(denom(Hcircuitideal)=0,s);
```

$$-\frac{1}{C\,r}$$

Evaluare numerica:

```
> H:=eval(Hideal,[ C=22*1E-9, R=10^3]);Hc:=eval(Hcircuitideal,[
C=22*1E-9, R=10^3, r=100]);PZ[numeric](Hc,s);
```

$$H := -.000022000\, s$$

$$Hc := -.000022000\,\frac{s}{1 + .2200\;10^{-5}\, s}$$

$$\begin{bmatrix} z1 & 0. \\ p1 & -454500. \end{bmatrix}$$

Polii functiei de transfer:

```
> Bode[castig](H);Bode[faza](H);
```




```
> Bode[polara](H);Bode[polara](Hc);
```



Interpretarea functiei de transfer:

```
> plot3d(abs(eval(H,s=sigma+I*omega)),sigma=-10^6..10^6,omega=-10^6..10^6,axes=normal,title="Reprezentarea in spatiu a modulului f.d.t.");
plot3d(abs(eval(Hc,s=sigma+I*omega)),sigma=-10^6..10^6,omega=-10^6..10^6,axes=normal,title="Reprezentarea in spatiu a modulului f.d.t.");
```





Functia de transfer in regim permanent H(j?)

```
> Hoideal:=subs(s=I*omega,Hideal);Hocircuitideal:=subs(s=I*omega,Hcircuitideal);
```

$$Hoideal := -I\,\omega\,C\,R$$

$$Hocircuitideal := \frac{-I\,\omega\,C\,R}{1+I\,\omega\,C\,r}$$

```
> assume(R,positive):assume(r,positive):assume(C,positive):
> abs_H0:=abs(limit(Hocircuitideal,omega=0));arg_H0:=argument(limit(Hocircuitideal,omega=0));
abs_Hinf:=abs(limit(Hocircuitideal,omega=infinity));arg_Hinf:=argument(limit(Hocircuitideal,omega=infinity));
abs_Halpha:=abs(eval(Hocircuitideal,omega=1/(r*C)));arg_Halpha:=argument(eval(Hocircuitideal,omega=1/(r*C)));alpha=eval(1/(r*C),[R=10^3, C=22*1E-9, r=100]);
```

$$abs\_H0 := 0,\quad arg\_H0 := 0$$

$$abs\_Hinf := \frac{R\sim}{r\sim},\quad arg\_Hinf := \pi$$

$$abs\_Halpha := \frac{1}{2}\frac{\sqrt{2}\,R\sim}{r\sim},\quad arg\_Halpha := -\frac{3}{4}\pi$$

$$\alpha = 454545.4545$$

```
> plot( eval( [ abs(Hoideal), abs(Hocircuitideal) ] , [R=10^3, C=22*1E-9, r=100]), omega=-10^6..10^6,axes=normal,title="Reprezentarea modulului f.d.t. pentru regim permanent", view=[DEFAULT, -10..200]);
plot( eval( [ argument(Hoideal), argument(Hocircuitideal) ] , [R=10^3, C=22*1E-9, r=100]), omega=-10^6..10^6,axes=normal,title="Reprezentarea modulului f.d.t. pentru regim permanent");
```





Raspuns in regim permanent

```
> restart:
> libname:="c:/maple/SCSlib",libname:
> F:=table([dir=FOURIER,inv=inttrans[invfourier]]):
```

Pentru circuite liniare lucrând în regim permanent sinusoidal functia de transfer are semnificatia de ***amplificare generalizata.*** Un semnal $e(t)=A\cos(w_0 t + j)$ aplicat la intrarea circuitului liniar descris de functia de transfer H(s) se regaseste la iesire sub forma:

$$y(t)= A\left|H(s)\right|_{s=jw_0}\cos(w_0 t + j + arg\,H(s)_{s=jw_0})$$

adica amplificat cu modulul functiei de transfer la frecventa w_0 si defazat suplimentar cu argumentul functiei de transfer la aceasi frecventa w_0.

Pentru un semnal de intrare format prin sumarea unui numar de semnale sinusoidale:

$$e(t)=A_0+\sum_{k=1}^{N} A_k\cos(w_{0k}t + j_k)$$

iesirea se poate calcula pe baza proprietatii de liniaritate:

$$y(t)= A_0 H(j0)+\sum_{k=1}^{N} A_k\left|H(jw_{0k})\right|\cos(w_{0k}t + j_k + \arg H(jw_{0k}))$$

Obs: Derivatorul are in IF amplificare teoretica infinita. Realizat practic un astfel de circuit nu functioneaza. El practic deriveaza o componenta de inalta frecventa parazita si se satureaza. Schema a doua este o varianta de realizare practica in care amplificarea la IF a fost limitata. Schema a doua se comporta ca derivator pentru frecvente mult mai mici decit $\alpha=\frac{1}{rC}$.

```
> Hs:=-(s*alpha1)/(alpha*(s+alpha1));
```

$$Hs := -\frac{s\,\alpha 1}{\alpha\,(s+\alpha 1)}$$

```
> Homega:=subs(s=I*omega,Hs);
```

$$Homega := \frac{-I\,\omega\,\alpha 1}{\alpha\,(I\,\omega + \alpha 1)}$$

Atenuarea in cc este:

```
> limit(Homega,omega=0);
```

$$0$$

Raspunsul la semnal armonic

In acest caz expresia excitatia $e(t)$ este de forma:

```
> e:=A0*cos(w*t);
```

$$e := A0\cos(w\,t)$$

Transformata Fourier a excitatie $e(t)$ este:

```
> E:=F[dir](e,t,omega);
```

$$E := A0\,\pi\,\mathrm{Dirac}(-\omega + w) + A0\,\pi\,\mathrm{Dirac}(\omega + w)$$

Transformata Fourier a excitatie $y(t)$ este:

```
> Y:=Homega*E;
```

$$Y := \frac{-I\,\omega\,\alpha 1\,(A0\,\pi\,\mathrm{Dirac}(-\omega + w) + A0\,\pi\,\mathrm{Dirac}(\omega + w))}{\alpha\,(I\,\omega + \alpha 1)}$$

Raspunsul $y(t)$ al circuitului la excitatia $e(t)$ este:

```
> y:=simplify(normal(convert(F[inv](Y,omega,t),trig),expanded));
```

$$y := -\frac{\alpha 1\,A0\,w\,(w\cos(w\,t) - \alpha 1\sin(w\,t))}{\alpha\,(w^2 + \alpha 1^2)}$$

La acelasi rezultat se putea ajunge urmarind pasii de calcul:

$$H(jw) = -\frac{a_1}{a}\frac{jw}{jw + a_1},\quad a_1 = \frac{1}{rC}, a = \frac{1}{RC}$$

Modulul si faza functiei de transfer:

$$\begin{cases} |H(jw)| = \dfrac{a_1}{a}\dfrac{w}{\sqrt{w^2 + (a_1)^2}} \\ \arg(H(jw)) = -\dfrac{p}{2} + arctg\left(\dfrac{a_1}{w}\right) \end{cases}$$

Forma generala a raspunsului permanent:

$$y(t) = A_0\,|H(jw)|\cos(wt + \arg(H(jw)))$$

Raspunsul permanent este:

$$y(t) = A_0\frac{a_1}{a}\frac{w}{\sqrt{w^2 + (a_1)^2}}\cos\left(wt - \frac{p}{2} + arctg\left(\frac{a_1}{w}\right)\right)$$

Raspunsul la semnal armonic de frecventa egala cu frecventa polului

```
> e1:=eval(e,w=alpha);
```

$$e1 := A0\cos(\alpha\,t)$$

```
> y1:=simplify(eval(y,w=alpha));
```

$$y1 := -\frac{\alpha 1\,A0\,(\alpha\cos(\alpha\,t) - \alpha 1\sin(\alpha\,t))}{\alpha^2 + \alpha 1^2}$$

Obs: Semnalul de iesire este defazat fata de intrare cu $-\frac{\pi}{4}$ si atenuat cu $\frac{1}{\sqrt{2}}$ in raport cu atenuarea din banda.

Raspunsul la semnal armonic de frecventa joasa

```
> e;
```

$$A0\cos(w\,t)$$

```
> y;
```

$$-\frac{\alpha 1\,A0\,w\,(w\cos(w\,t)-\alpha 1\sin(w\,t))}{\alpha\,(w^2+\alpha 1^2)}$$

```
> limit(y,alpha1=infinity);
```

$$\frac{A0\,w\sin(w\,t)}{\alpha}$$

Obs : iesirea este derivata intrarii!

Raspuns de regim tranzitoriu

```
> restart:with(inttrans):
> libname:="c:/maple/SCSlib",libname:
> L:=table([dir=inttrans[laplace],inv=inttrans[invlaplace]]):
> 
assume(_alpha,positive):assume(_omega,positive):assume(_tau,posi
tive):assume(_T,positive):
```

Consideram circuitul liniar cu AO pentru care am determinat functia de transfer $H(s)$. Excitatia este $e(t)$ si raspunsul este tensiunea $y(t)$. In situatia in care semnalul $e(t)$ este cauzal, circuitul functioneaza in regim tranzitoriu. In studiul comportarii de regim tranzitoriu se utilizeaza *transformata Laplace*.Vom nota cu $Y(s)$ transformata Laplace a semnalului $y(t)$ si cu $E(s)$ transformata Laplace a semnalului excitatie $e(t)$. Metodologia de calcul a raspunsului $y(t)$ la excitatia $e(t)$ pentru un circuitui este urmatoarea:

- Determinarea lui $E(s)$ din $e(t)$, folosind *transformata Laplace directa.*
- Determinarea lui $Y(s)$, folosind relatia $Y(s)=H(s)\,E(s)$.
- Determinarea lui $y(t)$ din $Y(s)$, folosind *transformata Laplace inversa.*

Pentru circuit functia de transfer este:

```
> Hs :=-(s*alpha1)/(alpha*(s+alpha1));
```

$$Hs := -\frac{s\,\alpha 1}{\alpha\,(s+\alpha 1)}$$

Raspunsul la semnal treapta

In acest caz expresia excitatia $e(t)$ este de forma:

```
> e:=A0*Heaviside(t);
```

$$e := A0\,\mathrm{Heaviside}(t)$$

- Transformata Laplace a excitatie $e(t)$ este:

```
> E:=L[dir](e,t,s);
```

$$E := \frac{A0}{s}$$

- Transformata Laplace a excitatie $y(t)$ este:

```
> Y:=Hs*E;
```

$$Y := -\frac{\alpha 1\, A0}{\alpha\,(s+\alpha 1)}$$

- Raspunsul $y(t)$ al circuitului la excitatia $e(t)$ este:

```
> y:=L[inv](Y,s,t)*Heaviside(t);
```

$$y := -\frac{\alpha 1\, A0\, \mathbf{e}^{(-\alpha 1\, t)}\, \text{Heaviside}(t)}{\alpha}$$

Se reprezita semnalele de intrare si iesire pentru valorile din schema ale rezistentelor si ale condensatorului:

```
> plot(eval([e,subs([alpha=1/(R*C), alpha1=1/(r*C)],y)],
[A0=1,C=22*1E-9, R=10^3, r=100]), t=-0.0001..0.0001,numpoints =
200,thickness=1,axes=box,title="Raspunsul la excitatie data a
circuitului cu AO",labels=["t","e(t),y(t)"]);
```



Raspunsul la semnal dreptunghiular

In acest caz expresia excitatia $e(t)$ este de forma:

```
> e:=A0*(Heaviside(t)-Heaviside(t-tau));
```

$$e := A0\,(\text{Heaviside}(t) - \text{Heaviside}(t-\tau))$$

- Transformata Laplace a excitatie $e(t)$ este:

```
> E:=subs(_tau=tau,L[dir](subs(tau=_tau,e),t,s));
```

$$E := A0\left(\frac{1}{s} - \frac{\mathbf{e}^{(-s\,\tau)}}{s}\right)$$

- Transformata Laplace a excitatie $y(t)$ este:

```
> Y:=Hs*E;
```

$$Y := -\frac{s\,\alpha 1\, A0\left(\frac{1}{s} - \frac{\mathbf{e}^{(-s\,\tau)}}{s}\right)}{\alpha\,(s+\alpha 1)}$$

- Raspunsul $y(t)$ al circuitului la excitatia $e(t)$ este:

```
> y:=L[inv](Y,s,t)*Heaviside(t);
```

$$y := -\frac{\alpha 1\, A0\,(-\mathbf{e}^{(\alpha 1\,(-t+\tau))}\, \text{Heaviside}(t-\tau) + \mathbf{e}^{(-\alpha 1\, t)})\, \text{Heaviside}(t)}{\alpha}$$

Se reprezita semnalele de intrare si iesire pentru valorile din schema ale rezistentelor si ale condensatorului:

```
> plot(eval([e,subs([alpha=1/(R*C), alpha1=1/(r*C)],y)],
[A0=1,tau = 0.0001, C=22*1E-9, R=10^3, r=100]), t=-
```

```
0.0002..0.0002,numpoints =
200,thickness=1,axes=box,title="Raspunsul la excitatie data a
circuitului cu AO",labels=["t","e(t),y(t)"]);
```



Raspunsul la succesiune de pulsuri dreptunghiulare, bipolar - simetrice

```
> N:=5:
```

In acest caz expresia excitatia $e(t)$ este de forma:

```
> e:=(-A0/2+A0*sum(Heaviside(t-n*T)-Heaviside(t-tau-n*T),n=0..N-
1))*Heaviside(t):
```

- Transformata Laplace a excitatie $e(t)$ este:

```
> E:=subs([_tau=tau,_T=T],L[dir](subs([T=_T,tau=_tau],e),t,s)):
```

- Transformata Laplace a excitatie $y(t)$ este:

```
> Y:=Hs*E:
```

- Raspunsul $y(t)$ al circuitului la excitatia $e(t)$ este:

```
>y:=subs([_tau=tau,_T=T,_alpha=alpha],L[inv](subs([T=_T,tau=_tau,
alpha=_alpha],Y),s,t))*Heaviside(t):
```

Se reprezita semnalele de intrare si iesire pentru valorile din schema ale rezistentelor si ale condensatorului:

```
> plot(eval([e,subs([alpha=1/(R*C), alpha1=1/(r*C)],y)], [A0=1,T
=0.0002,tau = 0.0001, C=22*1E-9, R=10^3, r=100]), t=-
0.0001..0.001,numpoints=1000, thickness=1,
axes=box,title="Raspunsul la excitatia data a circuitului cu
AO",labels=["t","e(t),y(t)"]);
```



Comportarea derivatorului

Comportare ca derivator

Se modifica frecventa semnalului dreptunghiular (perioda T) cu pastrarea unui factor de umplere de 50%. Se observa regimul tranzitoriu :

```
> plot(eval([e,subs([alpha=1/(R*C), alpha1=1/(r*C)],y)], [A0=1,T
```

```
=0.00005,tau = 0.000025, C=22*1E-9, R=10^3, r=100]), t=-
0.00001..0.00023,numpoints =
1000,thickness=1,axes=box,title="Raspunsul la excitatia data a
circuitului cu AO",labels=["t","e(t),y(t)"]);
```




```
> plot(eval([e,subs([alpha=1/(R*C), alpha1=1/(r*C)],y)], [A0=1,T
=0.0006,tau = 0.0003, C=22*1E-9, R=10^3, r=100]),
t=0.0011..0.0019,numpoints =
1000,thickness=1,axes=box,title="Raspunsul la excitatia data a
circuitului cu AO",labels=["t","e(t),y(t)"]);
```



Analiza PSPICE

Descriere PSpice:

```
*circuit
VIN 1 0 AC 1
VCC+ VCC+ 0 10V
VCC- VCC- 0 -10V
C1 1 21 22N
R1 21 31 1K
XOPAMP1 0 21 VCC+ VCC- 31 uA741
C2 20 22 22N
R2 22 32 1K
Rcom 1 20 100
XOPAMP2 0 22 VCC+ VCC- 32 uA741
.LIB "opamp.lib"
.AC dec 100 0.1 1000Meg
.PROBE
.END
```

```
*circuit
VIN 1 0 pulse(-1V 1V 0.3mS 1us 2us 0.3ms 0.602ms)
VCC+ VCC+ 0 10V
VCC- VCC- 0 -10V
C1 1 21 22N
R1 21 31 1K
XOPAMP1 0 21 VCC+ VCC- 31 uA741
C2 20 22 22N
R2 22 32 1K
Rcom 1 20 100
XOPAMP2 0 22 VCC+ VCC- 32 uA741
.LIB "opamp.lib"
.TRAN 5us 150ms 0 10us
.PROBE
.END
```

Diagrama Bode de modul si faza pentru cele doua circuite:



Functia pondere



Comportare de derivator